

Graphic Utilities Source Code Listings

```

001 ; GCLS -- Clear graphics screen
002 ;
003 PSECT 0F000H
004 GCLS PUSH HL ;Save registers
005 PUSH DE
006 PUSH BC
007 CALL INITG
008 LD A, INCY ;Set graphics status:
009 OUT (STATUS), A ; Graphics off, waits off, inc Y
010 XOR A
011 OUT (X), A ;Set X & Y address to 0
012 OUT (Y), A
013 LD B, 80 ;80 X addresses
014 OUTER LD C, B
015 LD B, 239 ;239 Y addresses. 240th done after loop.
016 INNER OUT (WRITE), A ;Zero graphics memory
017 DJNZ INNER ;Go clear next Y
018 LD A, INCXY ;Set status to inc X & Y after write
019 OUT (STATUS), A
020 XOR A
021 OUT (WRITE), A ;and clear last (240th) Y address
022 OUT (Y), A ;Set Y back to zero
023 LD A, INCY ;Reset status to inc Y only
024 OUT (STATUS), A
025 XOR A
026 LD B, C
027 DJNZ OUTER ;Go clear next X
028 LD A, 0FFH ;Set status to graphics, waits, no incs.
029 OUT (STATUS), A
030 POP BC ;Restore registers
031 POP DE
032 POP HL
033 XOR A
034 RET ;All done. Go back to caller.
035 INCY EQU 70H
036 INCXY EQU 30H
037 X EQU 80H
038 Y EQU 81H
039 WRITE EQU 82H
040 STATUS EQU 83H
041 ;
042 ; INITG -- Initialize Model III Graphics Board
043 ;
044 INITG LD A, 10H
045 OUT (236), A ;Turn on port
046 LD BC, 15

```

```
Ø47      LD      HL,DATA
Ø48 LOOP  LD      A,B          ;Program CRTC chip for 8Ø by 24
Ø49      OUT     (136),A
Ø5Ø      LD      A,(HL)
Ø51      OUT     (137),A
Ø52      INC     HL
Ø53      INC     B
Ø54      LD      A,B
Ø55      CP      C
Ø56      JR      NZ,LOOP
Ø57      RET
Ø58 DATA  DEFB   99
Ø59      DEFB   8Ø
Ø6Ø      DEFB   85
Ø61      DEFB   8
Ø62      DEFB   25
Ø63      DEFB   4
Ø64      DEFB   24
Ø65      DEFB   24
Ø66      DEFB   Ø
Ø67      DEFB   9
Ø68      DEFB   Ø
Ø69      DEFB   Ø
Ø7Ø      DEFB   Ø
Ø71      DEFB   Ø
Ø72      DEFB   Ø
Ø73      DEFB   Ø
Ø74 ;
Ø75      END     GCLS
```

```

001 ; GRON -- Turn on graphics display with waits on
002 ;
003         PSECT 0F000H
004 GRON    CALL  INITG
005         LD     A,0FFH
006         OUT   (STATUS),A
007         XOR   A
008         RET
009 STATUS EQU 83H
010 ;
011 ; INITG -- Initialize Model III Graphics Board
012 ;
013 INITG   PUSH  HL
014         PUSH  BC
015         PUSH  AF
016         LD     A,10H
017         OUT   (236),A      ;Turn on port
018         LD     BC,15
019         LD     HL,DATA
020 LOOP    LD     A,B          ;Program CRTC chip for 80 by 24
021         OUT   (136),A
022         LD     A,(HL)
023         OUT   (137),A
024         INC   HL
025         INC   B
026         LD     A,B
027         CP    C
028         JR    NZ,LOOP
029         POP   AF
030         POP   BC
031         POP   HL
032         RET
033 DATA   DEFB  99
034         DEFB  80
035         DEFB  85
036         DEFB  8
037         DEFB  25
038         DEFB  4
039         DEFB  24
040         DEFB  24
041         DEFB  0
042         DEFB  9
043         DEFB  0
044         DEFB  0
045         DEFB  0
046         DEFB  0
047         DEFB  0
048         DEFB  0
049 ;

```

Ø5Ø

END

GRON

```

001 ; GROFF -- Turn graphics display off with waits off
002 ;
003         PSECT 0F000H
004 GROFF  CALL  INITG
005         LD    A,0FCH
006         OUT  (STATUS),A
007         XOR  A
008         RET
009 STATUS EQU  83H
010 ;
011 ; INITG -- Initialize Model III Graphics Board
012 ;
013 INITG  PUSH  HL
014         PUSH  BC
015         PUSH  AF
016         LD    A,10H
017         OUT  (236),A      ;Turn on port
018         LD    BC,15
019         LD    HL,DATA
020 LOOP   LD    A,B          ;Program CRTC chip for 80 by 24
021         OUT  (136),A
022         LD    A,(HL)
023         OUT  (137),A
024         INC  HL
025         INC  B
026         LD  A,B
027         CP  C
028         JR  NZ,LOOP
029         POP  AF
030         POP  BC
031         POP  HL
032         RET
033 DATA  DEFB  99
034         DEFB  80
035         DEFB  85
036         DEFB  8
037         DEFB  25
038         DEFB  4
039         DEFB  24
040         DEFB  24
041         DEFB  0
042         DEFB  9
043         DEFB  0
044         DEFB  0
045         DEFB  0
046         DEFB  0
047         DEFB  0
048         DEFB  0
049 ;

```

Ø5Ø

END

GROFF

```

001 ; GSAVE -- Save graphics display to disk
002 ;
003 PSECT 0F000H
004 GSAVE PUSH HL ;Save registers
005 PUSH DE
006 PUSH BC
007 PUSH IY
008 PUSH HL
009 CALL INITG
010 LD HL,DCBEE ;Zero DCB buffer
011 LD DE,DCBEE+1
012 LD BC,49
013 LD (HL),00H
014 LDIR
015 POP HL
016 LD A,0DH
017 CP (HL)
018 JP Z,ERROR ;Error if filespec not given
019 LD DE,DCBEE
020 CALL 441CH ;Move filespec to DCB
021 JP NZ,BOMB
022 LD HL,BUFFER
023 LD DE,DCBEE
024 LD B,0
025 CALL 4420H ;Open file
026 JP NZ,BOMB
027 XOR A
028 LD (OPNFLG),A ;Set flag: file is open
029 ;
030 LD A,0E3H ;status = inc X after read
031 OUT (STATUS),A
032 XOR A
033 OUT (X),A ;init X & Y to zero
034 OUT (Y),A
035 LD E,A ;counter for X values
036 LD D,80 ;80 X values
037 LD B,75 ;75 disk records for entire screen
038 NXTREC LD HL,BUFFER
039 LD C,B
040 LD B,0 ;256 bytes per record
041 NGRPH IN A,(GRAPH) ;Get next graphics byte
042 LD (HL),A ; and put in buffer
043 INC HL
044 INC E
045 LD A,E
046 CP D
047 JR NZ,EGRPH ;Same row?
048 XOR A
049 LD E,A

```

```

050      OUT      (X),A          ;Next row. Set X to zero
051      LD       A,(YPOS)
052      INC      A
053      LD       (YPOS),A
054      OUT      (Y),A
055 EGRPH DJNZ    NGRPH          ;Go get next graphics byte
056      PUSH    DE
057      LD       DE,DCBEE
058      CALL    4439H          ;Write disk record
059      POP     DE
060      JR      NZ,BOMB
061      LD       B,C
062      DJNZ   NXTREC          ;Go fill buffer for next record
063 ;
064 EXIT  CALL    CLOSE
065      LD       A,0FFH          ;Status = graphics, waits, no incs
066      OUT      (STATUS),A
067      POP     IY
068      POP     BC
069      POP     DE
070      POP     HL
071      LD       A,(EFLAG)
072      CP      0
073      RET
074 ;
075 ; Subroutines
076 ;
077 CLOSE LD      A,(OPNFLG)
078      OR      A
079      RET     NZ              ;Return if file not open
080      LD      DE,DCBEE
081      JP     4428H          ;Go close file
082 ;
083 ; Error exits
084 ;
085 ;
086 ERROR LD      A,47          ;Required Command Parameter Not Found
087 ;
088 BOMB  LD      (EFLAG),A
089      LD      B,A
090      LD      A,39
091      RST    8              ;Print "ERROR nn" message
092      JP     EXIT
093 ;
094 X     EQU     80H
095 Y     EQU     81H
096 GRAPH EQU     82H
097 STATUS EQU    83H
098 EFLAG DEFB   0

```



```
099 YPOS  DEFB  0
100 OPNFLG DEFB  1
101 DCBEE  DEFS  50
102 BUFFER DEFS  256
103 ;  INITG  --  Initialize Model III Graphics Board
104 ;
105 INITG  LD    A,10H
106         OUT  (236),A      ;Turn on port
107         LD    BC,15
108         LD    HL,DATA
109 LOOP   LD    A,B          ;Program CRTC chip for 80 by 24
110         OUT  (136),A
111         LD    A,(HL)
112         OUT  (137),A
113         INC  HL
114         INC  B
115         LD    A,B
116         CP   C
117         JR   NZ,LOOP
118         RET
119 DATA  DEFB  99
120         DEFB  80
121         DEFB  85
122         DEFB  8
123         DEFB  25
124         DEFB  4
125         DEFB  24
126         DEFB  24
127         DEFB  0
128         DEFB  9
129         DEFB  0
130         DEFB  0
131         DEFB  0
132         DEFB  0
133         DEFB  0
134         DEFB  0
135 ;
136         END  GSAVE
```

```

001 ; GLOAD -- Save graphics display to disk
002 ;
003 PSECT 0F000H
004 GLOAD PUSH HL ;Save registers
005 PUSH DE
006 PUSH BC
007 PUSH IY
008 PUSH HL
009 CALL INITG
010 LD HL,DCBEE ;Zero DCB buffer
011 LD DE,DCBEE+1
012 LD BC,49
013 LD (HL),H
014 LDIR
015 POP HL
016 LD A,0DH
017 CP (HL)
018 JR Z,ERROR
019 LD DE,DCBEE
020 CALL 441CH ;Move filespec to DCB
021 JR NZ,BOMB
022 LD HL,BUFFER
023 LD DE,DCBEE
024 LD B,0
025 CALL 4424H ;Open file
026 JP NZ,BOMB
027 XOR A
028 LD (OPNFLG),A ;Set flag: file is open
029 ;
030 LD A,0B3H ;status = inc X after write
031 OUT (STATUS),A
032 XOR A
033 OUT (X),A ;init X & Y to zero
034 OUT (Y),A
035 LD E,A ;counter for X values
036 LD D,80 ;80 X values
037 LD B,75 ;75 disk records for entire screen
038 NXTREC PUSH DE
039 LD DE,DCBEE
040 CALL 4436H ;Read record from disk
041 POP DE
042 JR NZ,BOMB
043 LD HL,BUFFER
044 LD C,B
045 LD B,0 ;256 bytes per record
046 NGRPH LD A,(HL)
047 OUT (GRAPH),A
048 INC HL
049 INC E

```

```

050      LD      A,E
051      CP      D
052      JR      NZ,EGRPH      ;Same row?
053      XOR     A
054      LD      E,A
055      OUT     (X),A          ;Next row. Set X to zero
056      LD      A,(YPOS)
057      INC     A
058      LD      (YPOS),A
059      OUT     (Y),A
060 EGRPH DJNZ   NGRPH          ;Go get next graphics byte
061      LD      B,C
062      DJNZ   NXTREC          ;Go read next disk record
063 ;
064 EXIT  CALL   CLOSE
065      LD      A,0FFH          ;Status = graphics, waits, no incs.
066      OUT     (STATUS),A
067      POP     IY
068      POP     BC
069      POP     DE
070      POP     HL
071      LD      A,(EFLAG)
072      CP      0
073      RET
074 ;
075 ; Subroutines
076 ;
077 CLOSE LD     A,(OPNFLAG)
078      OR      A
079      RET     NZ              ;Return if file not open
080      LD      DE,DCBEE
081      JP      4428H          ;Go close file
082 ;
083 ; Error exits
084 ;
085 ERROR LD     A,47            ;Required Command Parameter Not Found
086 ;
087 BOMB  LD     (EFLAG),A
088      LD      B,A
089      LD      A,39
090      RST     8              ;Print "ERROR nn" message
091      JP      EXIT
092 ;
093 X     EQU    80H
094 Y     EQU    81H
095 GRAPH EQU    82H
096 STATUS EQU   83H
097 EFLAG DEFB  0
098 YPOS  DEFB  0

```

```

099 OPNFLG DEFB 1
100 DCBEE DEFS 50
101 BUFFER DEFS 256
102 ;
103 ; INITG -- Initialize Model III Graphics Board
104 ;
105 INITG LD A,10H
106 OUT (236),A ;Turn on port
107 LD BC,15
108 LD HL,DATA
109 LOOP LD A,B ;Program CRTC chip for 80 by 24
110 OUT (136),A
111 LD A,(HL)
112 OUT (137),A
113 INC HL
114 INC B
115 LD A,B
116 CP C
117 JR NZ,LOOP
118 RET
119 DATA DEFB 99
120 DEFB 80
121 DEFB 85
122 DEFB 8
123 DEFB 25
124 DEFB 4
125 DEFB 24
126 DEFB 24
127 DEFB 0
128 DEFB 9
129 DEFB 0
130 DEFB 0
131 DEFB 0
132 DEFB 0
133 DEFB 0
134 DEFB 0
135 ;
136 END GLOAD

```

```

001 ; GPRINT -- Print graphics screen to graphics printer
002 ;
003 PSECT 0F000H
004 GPRINT PUSH HL ;Save registers
005 PUSH DE
006 PUSH BC
007 PUSH IX
008 CALL INITG
009 OR 0DBH
010 OUT (STATUS),A ;Output a Control byte to cause
011 CALL INITBF ; Y to automatically dec. on a read
012 ;
013 XOR A ;Set A to 0
014 OUT (X),A ;Initialize the X position
015 LD (BPOS),A ; " " bit position
016 LD (XLOC),A ; " " " location counter
017 LD HL,BGMode
018 LD B,1
019 LD C,0DH
020 CALL PRLINE ;Begin graphics print mode
021 ;
022 LOOP1 LD IX,BUFFER ;point IX at the printer buffer
023 LD B,240 ;go through a whole column of bytes
024 LD A,B ;Put value in A and decrement
025 DEC A ; so it can be put out as
026 OUT (Y),A ; the Y position
027 COLUMN LD HL,MASK ;point HL at the mask byte
028 IN A,(GRAPH) ;input a graphics byte
029 AND (HL) ;chop off all but proper bit
030 CALL PO,SET0 ;if result is odd parity set bit 0
031 ; otherwise bit A is 0
032 LD HL,BPOS ;point HL at the bit position
033 PUSH BC ;save register B (for DJNZ loop)
034 LD B,(HL) ;get count
035 INC B ;increment (in case it is 0)
036 DECJ DEC B ;move bit left BPOS number of times
037 JR Z,PAST ;if done, move on...
038 RLC A ;move bit left one position
039 JR DECJ ;repeat loop
040 PAST POP BC ;get loop counter back
041 OR (IX) ;merge A with byte of printer buffer
042 LD (IX),A ;put merged result in buffer
043 INC IX ;increment buffer pointer
044 DJNZ COLUMN ;continue loop
045 ;-----
046 LD A,7 ;See if BPOS has gotten to 8.
047 INC (HL) ; If it has (printer uses 7 bits)
048 CP (HL) ; print the buffer and reset
049 CALL Z,PRNDRS ; BPOS to 0

```

```

050 ;
051 LD HL,MASK ;After getting a vertical row of bits
052 RRC (HL) ; rotate the mask right one position
053 LD A,80H ;Check to see if its back to
054 CP (HL) ; it's original value, if not
055 JR NZ,LOOP1 ; go get another row of bits
056 LD A,(XLOC) ;If so, get X pos (to increment it)
057 CP 79 ;Check to see if we are at the end...
058 JP Z,BYE
059 INC A ;otherwise increment the X counter
060 LD (XLOC),A ;and store it back
061 OUT (X),A ;also update the port value
062 JR LOOP1 ;now go get another row of bits
063 ;-----
064 SET0 LD A,1 ;set A to binary 0000 0001
065 RET ; and return
066 ;
067 PRNDRS LD HL,BUFFER ;Set up the
068 LD B,240 ; PRLINE call and
069 LD C,0DH ; send the buffer
070 CALL PRLINE
071 XOR A ;clear A
072 LD (BPOS),A ;reset bit position counter
073 ;
074 INITBF LD HL,BUFFER ;Initialize the printer buffer
075 LD DE,BUFFER+1 ; with all 80H
076 LD BC,239
077 LD A,80H
078 LD (HL),A
079 LDIR
080 RET
081 ;-----
082 PRLINE EQU $ ;Print a line. HL==>line to print
083 LD A,(HL) ;B = # characters to print
084 INC HL ;C = EOL char (sent after line)
085 CALL 3BH ;HL, BC, AF, and DE used
086 DJNZ PRLINE
087 LD A,C
088 CALL 3BH
089 RET
090 ;-----
091 BYE CALL PRNDRS
092 LD HL,EGMODE
093 LD B,1
094 LD C,0DH
095 CALL PRLINE ;End graphics print mode
096 POP IX ;Restore registers
097 POP BC
098 POP DE

```

```

099      POP      HL
100      XOR      A
101      RET
102 X    EQU      80H
103 Y    EQU      81H
104 GRAPH EQU     82H
105 STATUS EQU    83H
106 MASK  DEFB    80H      ;Mask to use in extracting bits
107 BGMODE DEFB   12H      ;Control byte: start graphics mode
108 BUFFER DEFS   240      ;Printer data buffer
109 EGMODE DEFB   1EH      ;Control byte: end graphics mode
110 BPOS  DEFB    0        ;Bit position in printer buffer
111 XLOC  DEFB    0        ;Current X location value
112 ;
113 ;
114 ;  INITG --  Initialize Model III Graphics Board
115 ;
116 INITG  LD      A,10H
117         OUT    (236),A      ;Turn on port
118         LD      BC,15
119         LD      HL,DATA
120 LOOP   LD      A,B          ;Program CRTC chip for 80 by 24
121         OUT    (136),A
122         LD      A,(HL)
123         OUT    (137),A
124         INC    HL
125         INC    B
126         LD      A,B
127         CP      C
128         JR      NZ,LOOP
129         RET
130 DATA  DEFB    99
131         DEFB    80
132         DEFB    85
132         DEFB    8
133         DEFB    25
134         DEFB    4
135         DEFB    24
136         DEFB    24
137         DEFB    0
138         DEFB    9
139         DEFB    0
140         DEFB    0
141         DEFB    0
142         DEFB    0
143         DEFB    0
144         DEFB    0
145 ;
146         END    GPRINT

```



```

00001 ; GPRT2 -- Print graphics X horizontal
00002 ;
00003 PSECT 0F000H
00004 GPRT2 PUSH HL
00005 PUSH DE
00006 PUSH BC
00007 PUSH IX
00008 CALL INITG
00009 LD HL,BGMode ;Turn on graphics print mode
00010 LD B,1
00011 LD C,0DH
00012 CALL PRLINE
00013 LD C,0 ;Graphics Y address
00014 LD A,0E3H
00015 OUT (STATUS),A
00016 ;
00017 NEWLN PUSH BC
00018 LD HL,BUF ;Clear buffer
00019 LD DE,BUF+1
00020 LD BC,639
00021 LD A,80H
00022 LD (HL),A
00023 LDIR
00024 ;
00025 POP BC
00026 LD D,1 ;Bit in buf to set
00027 ;
00028 NEWRW LD A,C
00029 OUT (Y),A ;Update Y address
00030 INC C
00031 LD HL,BUF
00032 XOR A
00033 OUT (X),A ;Restart X address
00034 LD B,80 ;Get 80 graphics bytes
00035 ;
00036 BYTE PUSH BC ;Save Y & loop counter
00037 IN A,(GRAPH)
00038 LD C,A ;Save graphics byte in C
00039 LD E,80H ;Get bits left to right
00040 BIT LD A,C
00041 AND E
00042 JR Z,OFF1
00043 LD A,D
00044 OR (HL)
00045 LD (HL),A ;Set bit in buffer
00046 OFF1 INC HL ;Next buffer byte
00047 SRL E ;Next bit

```

```

00048      JR      NZ,BIT
00049      POP     BC
00050      DJNZ   BYTE
00051 ;
00052      LD      A,240
00053      CP      C           ;Last Y address?
00054      JR      Z,DONE
00055      SLA     D           ;Next bit in buffer
00056      JP      P,NEWRW
00057 ;
00058      CALL   PRINT       ;Print buffer
00059      JR      NEWLN
00060 ;
00061 DONE  CALL   PRINT
00062      LD      A,0FCH
00063      OUT    (STATUS),A
00064      LD      HL,EGMODE   ;Turn off graphics print
00065      LD      B,1
00066      LD      C,0DH
00067      CALL   PRLINE
00068      POP     IX
00069      POP     BC
00070      POP     DE
00071      POP     HL
00072      XOR     A
00073      RET
00074 ;
00075 PRINT  PUSH   BC
00076      LD      DE,0       ;Offset for print buffer
00077 PART  LD      HL,BUF
00078      ADD    HL,DE
00079      XOR     A
00080      CP      (HL)       ;End of buffer?
00081      JR      Z,EPRT
00082      LD      BC,(CTL)
00083      CALL   PRLINE
00084      LD      HL,214
00085      ADD    HL,DE       ;Next part of buffer
00086      EX     DE,HL
00087      JR      PART
00088 EPRT  POP     BC
00089      RET
00090 ;
00091 PRLINE  EQU    $       ;Print a line. HL==>line to print
00092      PUSH   DE
00093 PRL2  LD      A,(HL)    ;B = # characters to print
00094      INC    HL          ;C = EOL char (sent after line)
00095      CALL   3BH        ;HL, BC, AF, and DE used
00096      DJNZ   PRL2

```

```

00097      LD      A,C
00098      OR      A
00099      CALL   NZ,3BH
00100      POP     DE
00101      RET
00102 ;
00103 ;  INITG --  Initialize Model III Graphics Board
00104 ;
00105 INITG  LD      A,10H
00106      OUT     (236),A      ;Turn on port
00107      LD      BC,15
00108      LD      HL,DATA
00109 LOOP  LD      A,B      ;Program CRTC chip for 80 by 24
00110      OUT     (136),A
00111      LD      A,(HL)
00112      OUT     (137),A
00113      INC     HL
00114      INC     B
00115      LD      A,B
00116      CP      C
00117      JR      NZ,LOOP
00118      RET
00119 DATA  DEFB   99
00120      DEFB   80
00121      DEFB   85
00122      DEFB   8
00123      DEFB   25
00124      DEFB   4
00125      DEFB   24
00126      DEFB   24
00127      DEFB   0
00128      DEFB   9
00129      DEFB   0
00130      DEFB   0
00131      DEFB   0
00132      DEFB   0
00133      DEFB   0
00134      DEFB   0
00135 ;
00136 BGMODE DEFB   12H
00137 EGMODE DEFB   1EH
00138 CTL    DEFB   0      ;Print 214 char, followed by null
00139      DEFB   214
00140 BUF   DEFS   640
00141      DEFB   0      ;Filler
00142      DEFB   0DH    ;Carriage return
00143      DEFB   0      ;End of buffer signal
00144 X    EQU   80H
00145 Y    EQU   81H

```

```
00146 GRAPH EQU 82H
00147 STATUS EQU 83H
00148 ;
00149          END GPRT2
```

```

00001 ; GPRT3 -- Print graphics X horizontal double Y axis
00002 ;
00003 PSECT 0F000H
00004 GPRT3 PUSH HL
00005 PUSH DE
00006 PUSH BC
00007 PUSH IX
00008 CALL INITG
00009 LD HL,BGMode ;Turn on graphics print
00010 LD B,1
00011 LD C,0DH
00012 CALL PRLINE
00013 LD C,0 ;Graphics Y address
00014 LD A,0E3H
00015 OUT (STATUS),A
00016 LD D,3 ;Bit(s) in buf to set
00017 ;
00018 NEWLN PUSH BC
00019 PUSH DE
00020 LD HL,BUF ;Clear buffer
00021 LD DE,BUF+1
00022 LD BC,639
00023 LD A,80H
00024 LD (HL),A
00025 LDIR
00026 ;
00027 POP DE
00028 POP BC
00029 ;
00030 NEWRW LD A,C
00031 OUT (Y),A ;Update Y address
00032 LD A,40H
00033 CP D
00034 JR Z,NEWRL ;If printing row second time
00035 INC C ;Move to next row
00036 NEWRL LD HL,BUF
00037 XOR A
00038 OUT (X),A ;Restart X address
00039 LD B,80 ;Get 80 graphics bytes
00040 LD A,4
00041 CP D
00042 JR NZ,BYTE
00043 LD D,6
00044 ;
00045 BYTE PUSH BC ;Save Y & loop counter
00046 IN A,(GRAPH)
00047 LD C,A ;Save graphics byte in C
00048 LD E,80H ;Get bits left to right

```

TRS-80[®]

```

00049 BIT      LD      A,C
00050          AND      E
00051          JR      Z,OFF1
00052          LD      A,D
00053          OR      (HL)
00054          LD      (HL),A      ;Set bit in buffer
00055 OFF1     INC      HL          ;Next buffer byte
00056          SRL      E          ;Next bit
00057          JR      NZ,BIT
00058          POP      BC
00059          DJNZ    BYTE
00060 ;
00061          LD      A,240
00062          CP      C          ;Last Y address?
00063          JR      Z,DONE
00064          SLA      D          ;Next bit in buffer
00065          SLA      D
00066          JR      Z,ENDRW
00067          JP      P,NEWRW
00068          LD      A,7FH
00069          AND      D
00070          LD      D,A
00071          JR      NZ,NEWRW
00072          LD      D,3
00073          JR      ENDR2
00074 ;
00075 ENDRW    LD      D,1
00076 ENDR2    PUSH     DE
00077          CALL    PRINT      ;Print buffer
00078          POP      DE
00079          JR      NEWLN
00080 ;
00081 DONE     CALL    PRINT
00082          LD      A,0FCH
00083          OUT     (STATUS),A
00084          LD      HL,EGMODE    ;Turn off graphics print
00085          LD      B,1
00086          LD      C,0DH
00087          CALL    PRLINE
00088          POP      IX
00089          POP      BC
00090          POP      DE
00091          POP      HL
00092          XOR      A
00093          RET
00094 ;
00095 PRINT     PUSH     BC
00096          LD      DE,0          ;Offset for print buffer
00097 PART      LD      HL,BUF

```

Radio Shack[®]

```

00098      ADD    HL,DE
00099      XOR    A
00100      CP     (HL)          ;End of buffer?
00101      JR     Z,EPRT
00102      LD     C,(CTL)
00103      CALL  PRLINE
00104      LD     HL,214
00105      ADD    HL,DE          ;Next part of buffer
00106      EX     DE,HL
00107      JR     PART
00108 EPRT  POP    BC
00109      RET
00110 ;
00111 PRLINE EQU  $          ;Print a line. HL==>line to print
00112      PUSH  DE
00113 PRL2  LD     A,(HL)      ;B = # characters to print
00114      INC   HL            ;C = EOL char (sent after line)
00115      CALL  3BH          ;HL, BC, AF, and DE used
00116      DJNZ PRL2
00117      LD     A,C
00118      OR    A
00119      CALL  NZ,3BH
00120      POP   DE
00121      RET
00122 ;
00123 ;  INITG --  Initialize Model III Graphics Board
00124 ;
00125 INITG  LD     A,10H
00126      OUT   (236),A        ;Turn on port
00127      LD     BC,15
00128      LD     HL,DATA
00129 LOOP  LD     A,B          ;Program CRTC chip for 80 by 24
00130      OUT   (136),A
00131      LD     A,(HL)
00132      OUT   (137),A
00133      INC   HL
00134      INC   B
00135      LD     A,B
00136      CP    C
00137      JR     NZ,LOOP
00138      RET
00139 DATA  DEFB  99

```

```

00140      DEFB  80
00141      DEFB  85
00142      DEFB   8
00143      DEFB  25
00144      DEFB   4
00145      DEFB  24
00146      DEFB  24
00147      DEFB   0
00148      DEFB   9
00149      DEFB   0
00150      DEFB   0
00151      DEFB   0
00152      DEFB   0
00153      DEFB   0
00154      DEFB   0
00155      ;
00156  BGMODE  DEFB  12H
00157  EGMODE  DEFB  1EH
00158  CTL     DEFB   0
00159          DEFB  214      ;Print 214 char, followed by null
00160  BUF     DEFS  640
00161          DEFB   0      ;Filler
00162          DEFB  0DH     ;Carriage return
00163          DEFB   0      ;End of buffer signal
00164  X       EQU   80H
00165  Y       EQU   81H
00166  GRAPH   EQU   82H
00167  STATUS  EQU   83H
00168      ;
00169          END    GPRT3

```